
Introduction to VHDL-AMS
Presented by Scott Cooper



Introduction to System Modeling Using VHDL-AMS 2

Presentation Agenda
VHDL-AMS Overview
Here we will briefly define what VHDL-AMS is, and some concepts associated 
with it.

Electrical Analog Modeling
In this portion of the presentation, we concentrate on analog, or continuous-time, 
modeling concepts with VHDL-AMS.

Mixed-Signal Modeling
In this section, we discuss mixed-signal modeling techniques. 

Power Converter Design Example
The last part of the presentation will focus on a power converter design developed 
with VHDL-AMS models.

What is VHDL-AMS?
A mixed-signal modeling language based on VHDL 
(IEEE 1076-1993)
A strict superset of VHDL (IEEE 1076.1-1999)

AMS => Analog / Mixed Signal Extensions

Represents complex models directly
Non-linear Ordinary Differential-Algebraic Equations (DAEs)
Mixed Analog/Digital

Can also model non-electrical physical phenomena

VHDL-AMS Concepts
VHDL-AMS models are organized as entities and 
architectures
It has a concept of time, concurrent processes
It has a well-defined simulation cycle
It can model continuous and discontinuous behavior
Equations are solved using conservation laws
(e.g. KCL, Newton’s Laws)
It handles initial conditions, piecewise-defined
behavior, and so forth

Electrical Analog Modeling

VHDL-AMS Model Structure
VHDL-AMS models are typically comprised of two 
sections: an entity and an architecture.

Entity - Describes the model interface to the outside 
world
Architecture - Describes the function or behavior of 
the model

Entity - model interface

Pins “p1” and “p2” provide the interface between this 
model and the outside world.

The nature of these pins is defined in the model’s 
“Entity declaration.” 

Resistor Model (Entity Declaration)

entity resistor is
port (

terminal p1, p2 : electrical);
end entity resistor;

Pin Definitions:

p1, p2 - electrical pins

Resistor Model (Entity Explanation)

entity resistor is
port (

terminal p1, p2 :electrical);  
end entity resistor;

Entity declaration

Device port (pin)  

Port type:
Analog?
Digital?
Conserved?

Port Nature:
Electrical?
Mechanical?
Thermal?
…?

Entity/model name

Entity/model name

Port names

Architecture - model behavior

The architecture describes the behavior of the model.

i = v / res

In this case, the model behavior is governed by Ohm’s 
law, which relates current and voltage as:

Resistor Model (Architecture)

architecture ideal of resistor is
constant res : real := 10.0e3;
quantity v across i through p1 to p2;

begin -- architecture ideal
i == v / res;

end architecture ideal;

Characteristic Equation:

i = v / res

Resistor Model (Architecture Explanation)

architecture ideal of resistor is
constant res : real := 10.0e3;
quantity v across i through p1 to p2;

begin -- architecture ideal

i == v / res;
end architecture ideal;

Entity nameArchitecture name

Architecture name

Model behavior

Internal object 
declarations

Now that we’ve seen the overall structure of a VHDL-AMS 
model, let’s explore some elements of the model.

VHDL-AMS Object Types
There are six classes of “objects” in VHDL-AMS: 

Constants
Terminals
Quantities
Variables
Signals
Files

For analog modeling, constants, terminals, and 
quantities are routinely used

Constants
Data storage object for use in a model

constant res : real := 50.0;
Declares constant, res, of type real, and initializes it to 50.0. Since this 
constant is of type real, it must be assigned only real values, which 
must include a decimal point.

constant count : integer := 3;
Declares constant count, of type integer, and initializes it to 3. Since 
count is of type integer, it must be assigned only whole values, which 
must not include a decimal point.

constant td : time := 1 ns;
Declares constant td, of type time, and initializes it to 1 ns (1.0e-9 
seconds). 
Time is a special kind of constant, described next.

Predefined Physical Types
The constant time is a predefined physical type, so named because 
it represents a real world physical property. It can be either real or 
integer.
As a physical type, time values are specified with a value followed 
by a multiplier (separated with a space). Predefined time 
multipliers consist of the following:
- fs (femto-seconds)
- ps (pico-seconds)
- ns (nano-seconds)
- us (micro-seconds)
- ms (milli-seconds)
- sec (seconds)
- min (minutes)
- hr (hours)
Since td is of type time, it may only be assigned time values.

Constants (cont.)
Constants make models easier to understand and 
modify (as opposed to using literal values)

i == v/50.0; -- Poor modeling style
i == v/res; -- Good modeling style

Constant values cannot be changed during simulation

Terminals
Terminals represent continuous, conservative ports
in VHDL-AMS
Terminals have across (potential) and through (flow) 
aspects
Terminal types are referred to as “natures”
Example terminal natures (predefined):

electrical - voltage  across, current  through
translational – position across, force through
thermal – temperature across, power (or heat-flow) through
fluidic – pressure across, flow-rate through

Users can define custom terminal natures

Quantities: 3 Types

Free quantity - non-conservative analog object:
quantity pwr : real;

Branch quantity - analog object used for conservative 
energy systems:

quantity v across i through p1 to p2;

Source quantity - for frequency domain:
quantity spectral_src real spectrum mag, phase;

Source quantities will not be discussed in this course

Free Quantities
Free quantities can be used to represent non-conserved 
analog values. 
They are often used to clarify model descriptions, and provide the 
ability to view internal model waveforms. Free quantities are also 
used to describe signal-flow (block diagram) type models.

quantity internal_variable : real := 5.0;
– In this case, the quantity internal_variable is of type real, and is initialized 

to 5.0.

quantity power : real;
– In this case, the quantity power is declared as type real, and is initialized to 

the default (left-most) value for that type. The default value for type real is 
guaranteed to be no larger than -1.0e+38. Depending on how they are used, 
it is sometimes important to initialize quantities and avoid their default 
values.

Branch Quantities
Branch quantity
Branch quantities are analog objects used for conservative energy 
systems. For electrical systems, these quantities are used to access 
either the voltage or current, or both, of a terminal port. 

To illustrate branch quantities, consider the entity declaration for 
the resistor model discussed previously:

terminal p1, p2 : electrical;

An example of the branch quantity declaration syntax for these 
terminals is next:

Branch Quantities

quantity v across i through p1 to p2 ;

Quantity “v” refers to 
the across aspect of 
terminal ports p1 and p2

Quantity “i” refers to 
the through aspect of
terminal ports p1 and p2

“v” and “i” are defined with respect
to terminal ports p1 and p2

Recall the resistor entity declaration for ports p1 and p2 :
terminal p1, p2 : electrical;

Since p1 and p2 are declared as electrical ports, v will represent voltage, 
and i will represent current
Any name can be used for the quantities (not restricted to v and i)

Source Quantities
Source quantity

Source quantities are used for frequency and noise modeling. 
These are used only in sources when frequency domain analysis is
to be performed, and other models do not require them to perform
in this domain. A syntax example is given as:

quantity spectral_src real spectrum mag, phase ;

Generic Constants (“Generics”)
Allow models to be externally parameterized
Static objects can be defined as generics in the entity
of a model, rather than as constants in the architecture 
of a model
Allows the model to be used more “generically,” 
without having to modify the model itself. The model 
user just passes in a value to the model.

Resistor Model (Entity with Generic)

entity resistor is
generic (

res : real := 10.0e3);
port (

terminal p1, p2 : electrical);

end entity resistor;

Generic type

Generic name

Optional initializer

Value of generic can be initialized in the entity
declaration. This value will be over-written if
specified when the component is instantiated.

architecture ideal of resistor is
constant res : real := 10.0e3;
quantity v across i through p1 to p2;

begin -- architecture ideal
i == v / res;

end architecture ideal;

Resistor Model 
(Architecture with Generic)

Constant res no longer defined in architecture

Implicit Quantity Attributes (analog)
Useful predefined quantity attributes

Q’dot
Time derivative of quantity Q

v == L*i’dot;  -- v = L*di/dt
Q’integ
Time integral of quantity Q

v == (1/C)*i’integ + init;  -- v = (1/C) ∫i dt + k
Q’delayed(T)
Quantity Q delayed by time T

v_out == v_in’delayed(td);
… many more

Analog Modeling Examples



Introduction to System Modeling Using VHDL-AMS 28

Inductor Model (Entity)

use ieee.electrical_systems.all;
entity inductor is

generic (
ind : real);    -- inductance value

port (
terminal p1, p2 : electrical); 

end entity inductor;

Pin Definitions/Argument:

p1, p2 : electrical pins
ind : user supplied argument

i

+ v -

p1 p2

Inductor Model (Architecture)

architecture ideal of inductor is
quantity v across i through p1 to p2;

begin -- ideal architecture
v == ind * i’dot;

end architecture ideal;

Fundamental Equation:

dt
diindv =i

+ v -

p1 p2

Diode Model (Entity)

entity diode is
generic (

-- saturation current
Isat : current := 1.0e-14; 

port (
terminal p, n : electrical);

end entity diode;

Pin Definitions/Argument:

p, n : electrical pins
Isat : user supplied argument

Diode Model (Architecture)

architecture ideal of diode is
constant TempC : real := 27.0;

constant TempK : real := 273.0 + TempC;

constant vt : real := PHYS_K*TempK/PHYS_Q;
quantity v across i through p to n;

begin
i == Isat*(exp(v/vt)-1.0);

end architecture ideal;

Fundamental Equation:

)0.1(exp* −= vt
v

Isati

Op Amp Model (Entity)

entity opamp_3p is
generic (

a_ol: real := 100.0e3;
f_0dB: real := 1.0e6

);
port (

terminal in_pos: electrical;
terminal in_neg: electrical;
terminal output: electrical

);
end entity opamp_3p;

Pin Definitions/Argument:

in_pos, in_neg, output : electrical pins
a_ol, f_0dB : user supplied arguments

in_pos

in_neg

output

Op Amp Model (Architecture)

architecture default of opamp_3p is
constant f_3dB: real := f_0dB/a_ol;

constant w_3dB: real := math_2_pi*f_3dB;

constant num: real_vector := (0 => a_ol);

constant den: real_vector := (1.0, 1.0/w_3dB);
quantity v_in across in_pos to in_neg;

quantity v_out across i_out through output to ELECTRICAL_REF;

begin
v_out == v_in'ltf(num, den);

end architecture default;

Fundamental Equation:

dB

ol

s
avinvout

3

1
ω

+
=

in_pos

in_neg

output
v_in v_out

Incandescent Lamp 
An incandescent lamp converts electrical energy into thermal energy. 
From an electrical standpoint, the lamp filament acts as a temperature-dependent 
resistance.
From a thermal standpoint, current flows through this resistance, power is developed 
and thermally dissipated as a combination of thermal conductance, thermal 
capacitance, and radiation.

i

Electrical model governing power 

v

R v*i = power => heat flow

(Powerelectrical = Powerthermal)

Thermal model governing power (heat flow)

T

rth cth

hflow

Lamp Equations
We begin with the electrical model of the preceding figure, which consists of a 
temperature-dependent electrical resistance. The power dissipated by this 
resistance is determined as follows:

power = v*i
where the power is simply the product of the voltage (v) across the electrical 
resistance and the current (i) through it. The voltage across the electrical 
resistance can be determined using Ohm’s law as follows:

v = i*R
where R represents the electrical resistance at the given temperature. This 
resistance, in turn, can be calculated with the following formula:

R = RC*(1.0 + alpha*(T - TC))
where RC is the electrical resistance when the lamp is “cold,” TC is the 
unheated “cold” temperature of the filament, T is the actual filament 
temperature, and alpha is the resistive temperature coefficient of the filament. 

Lamp Equations
We now have the necessary information to calculate the temperature-
dependent electrical power as a function of filament temperature. The next 
task is to develop equations which describe how this power is thermally 
dissipated.

For the thermal capacitance component (cth), the governing equation is: 
hflowcap = cth*dT/dt

where the heat flow is the product of the time derivative of the filament 
temperature (T) and the thermal capacitance (cth). The thermal conductance 
(rth) component is formulated as follows:

hflowres = (T - TA)/rth

where the heat flow is the ratio of the delta temperature (actual temperature 
(T) minus ambient temperature (TA)), and the thermal resistance. The lamp 
will also dissipate heat in the form of electromagnetic radiation. 

Lamp Equations
The radiated heat flow increases as the fourth power of the object’s 
temperature, and may be described as follows:

hflowradiated = Ke*(T4 - TA
4)

where Ke is the radiated energy coefficient. We now have all the equations 
necessary to implement the incandescent lamp model. 
To summarize our approach, we are attempting to equate electrical power to 
thermal power (heat flow), as follows:

Electrical: power = v*i
and

Thermal: hflow = hflowcap + hflowres + hflowradiated

these two equations may be equated by the following relationship:
Electrical/thermal: power = hflow 

Developing the Lamp Model
Although it is quite easy to develop simple models in an 
unstructured manner, more complex models benefit from a 
structured modeling approach. A recommended approach for 
analog modeling is:
1. Determine the model’s characteristic relationships for internal 

and external variables
2. Implement these relationships as simultaneous statements in 

VHDL-AMS
3. Declare appropriate objects to support the simultaneous 

statements

Incandescent Lamp (Architecture)

begin
r_temp == r_cold*(1.0 + alpha*(temp_fil - temp_cold_K));
v == i*r_temp;
hflow == v*i; -- Electrical power = heat flow
hflow == cth*temp_fil'dot + ke*SIGN(temp_fil - temp_amb_K)*(temp_fil**4

- temp_amb_K**4) + (temp_fil - temp_amb_K)/rth;
-- Note: For alpha, cth and rth, temperatures specified in C or K will work since each represents a ratio,
-- for which only the change in temperature is significant, not its absolute offset.

end architecture dyn_therm;

First, express core
relationships as VHDL-AMS
simultaneous equations…

architecture dyn_therm of Lamp is
constant temp_amb_K : real := temp_amb + 273.18;
constant temp_cold_K : real := temp_cold + 273.18;
quantity v across i through p1 to p2;
quantity r_temp : resistance; -- Resistance at temp_fil [ohms]
quantity temp_fil : temperature; -- Filament temperature [K]
quantity hflow : heat_flow; -- Heat flow from filament [watts]

…then declare appropriate
objects to support the
simultaneous equations.

Incandescent Lamp (Entity)

entity Lamp is
generic (

r_cold    : resistance  := 0.2;     -- Filament resistance at temp_cold
temp_cold : temperature := 27.0; -- Calibration temperature [deg C]
alpha : real := 0.0045; -- Resistive temp coefficient [ohms/deg C]
ke : real := 0.85e-12; -- Radiation coefficient [watts/K^4]
rth : real := 400.0; -- Thermal conduction [deg C/watt]
cth : real := 0.25e-3; -- Thermal heat capacitance [joules/C]
temp_amb : temperature := 27.0); -- Ambient temperature [deg C]

port (terminal p1, p2 : electrical);
end entity Lamp;

Finally, define the model’s interface to the outside world.

Model Solvability
Analog models are solved by the simulator as 
simultaneous equations. When solving simultaneous 
equations, the number of equations must equal the 
number of unknowns to be solved.

To ensure the same number of equations and unknowns 
in a behavioral model, the following formula may be 
applied:

# equations = # free quantities 
+ # through quantities 
+ # quantity ports of mode out

Mixed-Signal Modeling

Mixed-Signal Introduction
In this section, we combine the analog and digital 
modeling capabilities of VHDL-AMS. 

An overview of A/D and D/A conversion techniques 
will be given next, followed by specific model 
examples.

Analog to Digital
The ‘above attribute is used to convert an analog (continuous) quantity 
into a digital (discontinuous) signal, by detecting an analog threshold 
crossing. The syntax is as follows:

Q’above(threshold);

Where Q is the analog quantity to be converted, and threshold is the 
analog threshold level. 

This statement returns a boolean ‘true’ if quantity Q passes from 
below to above the threshold level; it returns a boolean ‘false’ if 
quantity Q passes from above to below the threshold level.

Digital to Analog
There are two primary methods for converting from digital signals to analog 
quantities. The first method involves using the ‘ramp attribute, as follows: 

Q == S’ramp(tr,tf);
Where Q is an analog quantity, S is a digital signal, and tr and tf are the rise 
and fall-times of Q at transition points.

When signal S changes value, quantity Q tracks this change, but transitions to 
it over a linear interval of tr or tf, depending on the direction of the change. 

The ‘ramp attribute also performs the function of restarting the analog solver at 
the discontinuous points when signal S is updated. This is a very important 
consideration for analog simulation so that the simulator does not get “lost” 
when encountering a discontinuity. 

Digital to Analog
The second method for D/A conversion can be used any time a 
quantity is updated as a function of a signal, as in:

Q == f(S);
break on S;

Where Q is an analog quantity, and f(S) is some function which returns 
a digital signal. 

In this case, if the ‘ramp attribute is not included in the statement, a 
break statement should be included to synchronize the analog quantity 
to the digital signal during state transitions. The break statement is 
used explicitly to accomplish what the ‘ramp attribute does implicitly, 
which is to guide a simulator through discontinuities.

Mixed Signal Model Examples

Analog to Digital Interface (Entity)

entity a2d is
generic (vthreshold : real := 2.0);    
port (d_output : out std_logic;

terminal a_input : electrical); 
end entity a2d;

Entity declaration can include both analog and digital ports.

Analog to Digital Interface (Architecture)

architecture behavioral of a2d is
quantity vin across a_input to electrical_ref;

begin
process (vin’above(vthreshold)) is
begin

if vin’above(vthreshold) then
d_output <= ‘1’;

else
d_output <= ‘0’;

end if;
end process;

end architecture behavioral;
-- Note that no simultaneous equations are required --

The ‘above Attribute
“Why use ‘above instead of > or < ?” 

‘above() generates an event exactly when the crossing occurs.  
– It is the only way to generate a signal to use in a break statement.

< or > do a comparison at each time-step, which may not fall on 
the exact crossing.

Must use “ not ‘above() ” to represent “ ‘below() ”, as 
‘below() is not part of the VHDL-AMS language.

Simple Switch and Entity

The purpose of this switch is to allow or prevent current flow between pins p1
and p2, depending on the value of sw_state. Ports p1 and p2 are electrical 
analog, and port sw_state is std_logic digital.

entity switch_dig_nogen is
port ( sw_state : in std_logic;

terminal p1, p2 : electrical );
end entity switch_dig_nogen;

Simple Switch Architecture
architecture ideal of switch_dig_nogen is

constant r_open : real := 10.0e3;
constant r_closed : real := 15.0e-3;
constant trans_time : real := 10.0e-6;
signal r_sig : resistance := r_open;
quantity v across i through p1 to p2;
quantity r : resistance;

begin
DetectState: process (sw_state)
begin
if (sw_state = ‘0’) then

r_sig <= r_open;
elsif (sw_state = ‘1’) then

r_sig <= r_closed;
end if;

end process DetectState;
r == r_sig’ramp(trans_time, trans_time);
v == r*i;

end architecture ideal;

Key Switch Attributes
‘event
The mechanism by which digital events are detected in 
VHDL-AMS. The switch uses this to detect when a new 
digital control signal is given.
‘ramp
Ensures that when switching from one value of r_sig to 
another, a reasonable amount of “switching time” is 
used.

The syntax used in the switch example is: 
q == S’ramp(tr, tf);

where q is a quantity (r), S is a signal (r_sig), and tr and tf are real values, 
representing the rise time and fall time respectively. 

Ideal Limiter Model (Entity)
A limiter model entity is shown below. Its function is to restrict the range of 
voltage levels which pass from input to output. In order to implement this 
behavior, the model selects between one of three simultaneous equations, 
depending on the level of the input voltage. 

entity limiter_ideal is
generic (

limit_high : real := 10.0; -- upper limit
limit_low : real := -10.0); -- lower limit

port (

terminal input: electrical;
terminal output: electrical);

end entity limiter_ideal ;

Ideal Limiter Model (Architecture)
architecture simple of limiter_ideal is

quantity vin across input to electrical_ref;
quantity vout across iout through output to electrical_ref;

begin
if vin’above(limit_high) use -- above upper limit, so limit output

vout == limit_high;
elsif not vin’above(limit_low) use -- below lower limit, so limit output

vout == limit_low;
else -- no limit exceeded, so pass input signal to output

vout == vin;
end use;
break on vin'above(limit_high), vin'above(limit_low);

end architecture simple;

Power Converter Example

Phase 1 – Averaged Model
Phase 1 allows both frequency and time domain exploration 
of the design. It simulates very quickly, and allows the 
overall control loop to be stabilized.

Averaged Model Stimulus and Results

Dynamic load perturbations

Input perturbations

Output voltage

Phase 2 – Ideal Switch/Diode Models
Phase 2 allows switching effects to be analyzed with 
relatively ideal switch and diode models. Drive voltages 
and currents can now be evaluated.

Averaged and Switched Results
Superimposed output voltages for 
averaged and switched designs

Phase 3 – MOSFET, IRR Diode and Stress 
Checks

Phase 3 includes a MOSFET switch, a diode model that 
includes IRR effects, as well as stress monitoring for the 
MOSFET and diode.

Averaged, Ideal Switch, MOSFET Tests
Superimposed output voltages for 
averaged, ideal switched and MOSFET 
switch-based designs

Stress Indicators in Waveform Viewer
Boolean indicators in the Waveform Viewer that show if 
any stress measures have been violated. 

Analyzing Causes of Stress (IRR Violated)

If MOSFET switches too 
fast (RGATE = 1k), IRR 
specification is violated. 

Analyzing Causes of Stress (MOSFET Power)

If MOSFET switches too 
slow (RGATE = 3k), its power 
rating is violated. 

Test IRR Diode
Test circuit for the diode with 
reverse-recovery effects 
modeled. 

IRR Diode Test Results

Test results for the diode 
with reverse-recovery 
effects modeled. 

VHDL-AMS Reference Material
How to Model Mechatronic Systems Using VHDL-AMS is the first booklet in the 
SystemVision Technology Series. This booklet serves as the foundation for this 
modeling course. It is available from the SystemVision Welcome Screen, and from 
Mentor Graphics at the SystemVision website.

The System Designer's Guide to VHDL-AMS (P. J. Ashenden, G. D. Peterson, D. 
A. Teegarden - ISBN 1-55860-749-8, published by Morgan-Kaufman Publishers, 
2002) is a comprehensive textbook for the VHDL-AMS modeling language. 

The Designer’s Guide to Analog & Mixed-Signal Modeling (R. S. Cooper - ISBN 
0-9705953-0-1, published by Avant!, 2001) includes numerous model examples in 
both the VHDL-AMS and MAST modeling languages.

The VHDL-AMS Quick Reference Guide offers a summary of many VHDL-AMS 
language features and syntax. It is accessed from within SystemVision: Help > 
Manuals > VHDL-AMS Quick Reference.

Thank You!


